

```

MODEL:
! A Simple Transportation Problem ;
! This version checks if there is a feasible solution, and
! if not, calls the DEBUG analysis;
! Keywords: CALC, Debug, Distribution, Infeasibility, Infeasible,
! LINGO, Shipping, Transportation;
SETS:
SOURCE          : CAP;
CUSTOMER        : DEM;
ROUTE( SOURCE, CUSTOMER)      : SCOST, SHIPT;
ENDSETS
! Here are the parameters;
DATA:
SOURCE = BILOXI LAX TAMPA; ! Names of supply points;
CAP    = 30, 25, 15 ; ! Their capacities;
! Names of demand points;
CUSTOMER = AURORA CASPER TONOPAH LUDLOW;
DEM      = 24, 20, 95, 21 ; ! Notice, the 95 for TONOPAH seems
suspicious;
SCOST =
        6, 2, 6, 7,
        4, 9, 5, 3,
        8, 8, 1, 5 ;

ENDDATA
!-----;

SUBMODEL TRANIT:
! Variables:
SHIPT( i, j) = amount shipped from source i to destination j;
! The objective;
MIN = OBJ;
OBJ = @SUM( ROUTE: SCOST * SHIPT); ! Cost of shipments;
! The demand constraints;
@FOR( CUSTOMER( J):
[DMRO] @SUM( SOURCE( I): SHIPT( I, J)) = DEM( J));
! The supply constraints;
@FOR( SOURCE( I):
[CPRO] @SUM( CUSTOMER( J): SHIPT( I, J)) <= CAP( I));
ENDSUBMODEL
CALC:
@SET( 'TERSEO',3); ! Output level (0:verbose, 1:terse, 2:only errors, 3:none);
@SOLVE( TRANIT);
ISTAT = @STATUS(); ! 0: Optimal to tolerance. 1: infeasible, 2: unbounded,
3: undetermined, 4: Feasible, 5: Infeasible/unbounded in
preprocessor,
6: Local optimum, 7: locally infeasible, 8: Objective cutoff
reached,
9: numeric error;
@IFC( ISTAT #EQ# 0:
@WRITE(' Optimal feasible solution found. Cost= ', OBJ, @NEWLINE(1));
! Display the shipments;
@FOR( ROUTE( I, J) | SHIPT( I, J) #GT# 0:
@WRITE(' Ship ', @FORMAT( SHIPT(I,J),'5.1f'),' from ',
@FORMAT( SOURCE(i),'7s'),' to ', @FORMAT( CUSTOMER(j),'7s'), @NEWLINE(1));
);
@ELSE
@WRITE(' Model is infeasible, do DEBUG analysis', @NEWLINE( 1));
@SET( 'TERSEO',0); ! Output level (0:verbose, 1:terse, 2:only errors, 3:none);
! If using LINGO interactively, can execute DEBUG by clicking on:
Solver -> Debug ;
@DEBUG( TRANIT); ! Find a small set of rows that are infeasible;
);
ENDCALC
END

```